

# Computergestuetzte Mathematik zur Analysis

## Lektion 3 (Integration und Differentiation)

### Summen und Reihen (Wdh.)

```
> restart;
```

```
> sum(j,j=0..n);
```

$$\frac{1}{2} (n+1)^2 - \frac{1}{2} n - \frac{1}{2} \quad (1.1)$$

```
> sum(1/j^4, j = 1 .. infinity);
```

$$\frac{1}{90} \pi^4 \quad (1.2)$$

### Grenzwerte / Limes

```
> limit((1+z/n)^n,n=infinity);
```

$$e^z \quad (2.1)$$

```
> limit( (x^2+2*x-3)/(x-1),x=1);
```

$$4 \quad (2.2)$$

```
> limit( sum(1/j,j=1..n)-ln(n),n=infinity);
```

### Integration und Differentiation

```
> f := 1/(1+t^2);
```

$$f := \frac{1}{t^2 + 1} \quad (3.1)$$

```
> If := int(f, t);
```

$$If := \arctan(t) \quad (3.2)$$

```
> diff(If,t);
```

$$\frac{1}{t^2 + 1} \quad (3.3)$$

```
> int(f,t=a..b);
```

$$-\arctan(a) + \arctan(b) \quad (3.4)$$

```
> ff := x -> 1/(1+x^2);
```

$$ff := x \rightarrow \frac{1}{x^2 + 1} \quad (3.5)$$

```
> int(ff,x); # Ist das richtig ?
```

$$ff x \quad (3.6)$$

```
> int(ff(t),t);
```

$$\arctan(t) \quad (3.7)$$

```
> diff(ff,t);
```

$$0 \quad (3.8)$$

```
> diff(ff(y),y);
```

$$-\frac{2y}{(y^2+1)^2} \quad (3.9)$$

```
> restart;
```

```
> f := exp(Pi*I*x);
```

$$f := e^{I\pi x} \quad (3.10)$$

```
> d2f := diff(diff(f,x),x);
```

$$d2f := -\pi^2 e^{I\pi x} \quad (3.11)$$

```
> d2f := diff(f,x,x);
```

$$d2f := -\pi^2 e^{I\pi x} \quad (3.12)$$

```
> x$4;
```

$$x, x, x, x \quad (3.13)$$

```
> d4f := diff(f,x$4);
```

$$d4f := \pi^4 e^{I\pi x} \quad (3.14)$$

```
> Diff(f,x$3);
```

$$\frac{d^3}{dx^3} e^{I\pi x} \quad (3.15)$$

Kettenregel:

```
> f:='f':
```

```
> g:='g':
```

```
> diff(f(g(x)),x);
```

$$D(f)(g(x)) \left( \frac{d}{dx} g(x) \right) \quad (3.16)$$

Produktregel:

```
> diff(f(x)*g(x),x);
```

$$\left( \frac{d}{dx} f(x) \right) g(x) + f(x) \left( \frac{d}{dx} g(x) \right) \quad (3.17)$$

Quotientenregel:

```
> # ÜA :)
```

## ▼ Traege Operatoren

```
> s:= Sum(1/j^2,j=1..infinity);
```

$$S := \sum_{j=1}^{\infty} \frac{1}{j^2} \quad (4.1)$$

> value(S);

$$\frac{1}{6} \pi^2 \quad (4.2)$$

> f := sin(x)/x;

$$f := \frac{\sin(x)}{x} \quad (4.3)$$

> Df := Diff(f,x);

$$Df := \frac{d}{dx} \left( \frac{\sin(x)}{x} \right) \quad (4.4)$$

> value(Df);

$$\frac{\cos(x)}{x} - \frac{\sin(x)}{x^2} \quad (4.5)$$

> If := Int(f,x=-1..1);

$$If := \int_{-1}^1 \frac{\sin(x)}{x} dx \quad (4.6)$$

> If := value(If);

$$If := 2 \operatorname{Si}(1) \quad (4.7)$$

> evalf(If);

$$1.892166141 \quad (4.8)$$

> Limit(Sum(1/j,j=1..n)-ln(n),n=infinity);

$$\lim_{n \rightarrow \infty} \left( \sum_{j=1}^n \frac{1}{j} - \ln(n) \right) \quad (4.9)$$

> value((4.9));

$$\gamma \quad (4.10)$$

## ▼ Differentiation II

> f := 1/(1+t^2);  
ff := t -> 1/(1+t^2);

$$f := \frac{1}{t^2 + 1}$$

$$ff := t \rightarrow \frac{1}{t^2 + 1} \quad (5.1)$$

> value(Diff(ff(t),t));

$$(5.2)$$

$$-\frac{2t}{(t^2+1)^2} \quad (5.2)$$

> diff(f,t);

$$-\frac{2t}{(t^2+1)^2} \quad (5.3)$$

> unapply(value(Diff(ff(t),t)),t);  
D(ff); # :)

$$t \rightarrow -\frac{2t}{(t^2+1)^2}$$

$$t \rightarrow -\frac{2t}{(t^2+1)^2} \quad (5.4)$$

> D(arctan); #Funktionsschreibweise

$$z \rightarrow \frac{1}{z^2+1} \quad (5.5)$$

> f:=exp@sin; #Verkettung

$$f := \exp @ \sin \quad (5.6)$$

> f(x);

$$e^{\sin(x)} \quad (5.7)$$

> g:=exp@exp@exp;

$$g := \exp^{(3)} \quad (5.8)$$

> g(x);

$$\exp^{(3)}(x) \quad (5.9)$$

> expand(g(x));

$$e^{e^{e^x}} \quad (5.10)$$

> G:=exp@@3;

$$G := \exp^{(3)} \quad (5.11)$$

> G-g;

$$0 \quad (5.12)$$

> (D@@2)(f);

$$\exp @ \sin \cos^2 - \exp @ \sin \sin \quad (5.13)$$

> expand((5.13)(x));

$$e^{\sin(x)} \cos(x)^2 - e^{\sin(x)} \sin(x) \quad (5.14)$$

> f:='f'; g:='g';

$$f := f$$

$$g := g \quad (5.15)$$

> D(f@g);

$$D(f) @ g D(g)$$

(5.16)

> expand((5.16)(x));

$$D(f) (g(x)) D(g)(x)$$

(5.17)

> D(f/g);

$$\frac{D(f)}{g} - \frac{fD(g)}{g^2}$$

(5.18)

> normal((5.18));

$$\frac{D(f) g - fD(g)}{g^2}$$

(5.19)